STATFINN

THE RIGHT WAY TO CONCLUSIONS

PK Plots with SGPLOT

Tiina Kirsilä

PhUSE Singe Day Event 2011
Copenhagen

Contents

Statistical Graphics Procedures

SG Procedures covered in this presentation

- SGPLOT
  - procedure creates a single-celled graph, with multiple plots overlaid within a single set of axes
- SGPANEL
  - procedure creates classification panels for one or more classification variables

SG Procedures not covered in this presentation

- SGSCATTER (creates panelled graphs with multiple scatter plots)
- SGRENDER (creates graphs from templates that are written in the Graph Template Language)
- SGDESIGN (creates graphical output based on a graph file that has been created by using the SAS/GRAPH ODS Graphics Designer application)

Statistical Graphics Procedures Syntax

SGPLOT

PROC SGPLOT <DATA= *input-data-set*> < option(s)>;

SCATTER X= *variable* Y= *variable* </*option(s)*>; /***Creates a scatter plot.** */

SERIES X= *variable* Y= *variable* </*option(s)*>; /***Creates a line plot.** */

VBOX *response-variable* </*option(s)*>; /***Creates a vertical box plot that shows the distribution of your data.** */

XAXIS <*option(s)*>;

YAXIS <*option(s)*>; /***specify the axis options for each plot axis** */

KEYLEGEND <"*name-1*" … "*name-n*"> </*option(s)*>; /***Adds a legend to the plot.** */

REFLINE *value(s)* </*option(s)*>; /***Creates a horizontal or vertical reference line.** */

SGPANEL

PROC SGPANEL <DATA= *input-data-set*> < option(s)>;

PANELBY *variable(s)*< |*option(s)*>;

.

.

Controlling graph appearance

Attributes

- Markers (symbol, color, size)
- Lines (pattern, color, thickness, transparency)

Axis attributes

- Labels
- Type (discrete, linear, log..)
- Tick values (0 to 10 by 1, format,..)

ODS styles and ODS graphics

ODS listing styles

- Multiple ODS styles available, which can also be modified

  PROC TEMPLATE; LIST STYLES; /***produces a list of the current SAS styles available** */

ODS graphics on

- Statement is needed if one whishes to change the size or format of the graph

Pharmacokinetics (PK)

What happens to the drug within the body?

PK is the science of measuring and interpreting the concentrations of the drug and metabolite in blood (or urine) after the drug has been dosed

PK parameters

C_{max} - peak concentration
T_{max} - time at which maximum concentration occurs
C_{last} - last measure concentration
T_{last} - time at which maximum concentration occurs
AUC_t - the area under the concentration-time curve from dosing to T_{last}
K_{el} - the elimination rate constant, represents the decrease in concentration per unit of time
AUC_{inf} - the area under the concentration-time curve extrapolated to infinity

Concentration data

Treatment	*Time (h)*	*Concentration (mg/ml)*	*Subject*	*sequence*	*visit*
R1	0.0	0.0	1	2	1
R1	0.5	19.4	1	2	1
R1	1.0	31.5	1	2	1
R1	1.5	37.4	1	2	1
R1	2.0	45.2	1	2	1
R1	2.5	44.5	1	2	1
R1	3.0	47.9	1	2	1
R1	4.0	49.8	1	2	1
R1	5.0	47.1	1	2	1
R1	6.0	48.4	1	2	1
R1	7.0	47.1	1	2	1
R1	8.0	40.5	1	2	1
R1	9.0	36.6	1	2	1
R1	10.0	31.7	1	2	1
R1	11.0	30.8	1	2	1
R1	12.0	25.6	1	2	1
R1	24.0	6.9	1	2	1

.

.

PK data

Subject	*Sequence*	*Treatment*	*Cmax*	*Tmax*	*Tlast*	*Clast*	*AUCt*	*Kel*	*AUCinf*	*%AUCextrap*
1	2	T	39.963	2.5	12	7.711	286.350	0.241	318.335	10.048
2	1	T	34.627	1.5	11	3.246	184.750	0.324	194.759	5.139
3	3	T	38.456	2.0	12	6.237	264.475	0.231	291.500	9.271
1	2	R1	49.783	4.0	24	6.901	664.725	0.113	726.001	8.440
2	1	R1	32.161	1.5	11	3.056	173.750	0.355	182.369	4.726
3	3	R1	34.203	2.0	12	4.898	225.375	0.233	246.427	8.543
1	2	R2	31.803	1.5	12	3.215	180.225	0.243	193.427	6.825
2	1	R2	35.608	2.0	12	7.928	239.450	0.115	308.578	22.402
3	3	R2	27.657	1.5	8	3.475	115.050	0.343	125.195	8.103
.										
.										
.										

Concentration plots

```
ods graphics on /width=12cm height=12cm noborder ;
ods listing image_dpi=300 style=listing;
title 'Overlaid individual concentration-time profiles';
proc sgplot data=concentration noautolegend ;
series x=t y=concen/group=order
lineattrs=(color=black pattern=1);
run;
```



Concentration plots

```
title 'Overlaid individual concentration-time profiles by treatment';
proc sgpanel data=concentration noautolegend ;
format trt $treat.;
panelby trt / novarname layout=columnlattice;
series x=t y=concen/group=order lineattrs=(color=black pattern=1) ;
run;
```



Concentration plots

```
proc sgpanel data=concentration;
panelby id /layout=panel;
series x=t y=concen/group=trt;
run;
```



Concentration plots

```
title 'Individual concentration-
time profiles by treatment';
proc sgpanel data=concentration;
panelby id /onepanel
           layout=panel;
series x=t y=concen/group=trt;
run;
```



Concentration plots

```
title 'Individual log-concentration-time profiles by treatment';
proc sgpanel data=concentration ;
where id LE 4;
panelby id /   layout=panel;
series x=t y=concen/group=trt;
rowaxis type=log;
colaxis type=time;
run;
```



Concentration plots

```
proc means data=concentration mean
std;
class trt t;
var concen;
output out=out1 mean = mean std=std
run;
.
.
title 'Mean(SD) concentration-time
profiles';
proc sgplot data=out1;
scatter Y=mean X=t /group=trt
yerrorlower=lower yerrorupper=upper
keylegend / location=inside
position=topright;
series Y=mean X=t /group=trt ;
run;
```



PK plots

```
title 'AUCt values by
treatment';
proc sgplot data=pk;
vbox auct / category=trt;
xaxis display=(nolabel);
yaxis label= 'AUCt';
run;
```



PK plots - concentration curve with regression line

```
data kel;
merge concentration pk;
by id visit;
if concen EQ . then delete;
proc sort data=kel; by order descending t; run;

data kel;
set kel;
by order;
if first.order then do;
   Npoints   = 0;
end;
  Npoints + 1 ;
run;

data kel;
set kel;
if t LT tmax then delete;
if Npoints GT n_used then selected=0;
if Npoints LE n_used then selected=1;
reg = exp((estimate + (-kel)*t)*selected);
if selected EQ 0 then reg = .;
if selected EQ 1 then concen2=concen;
run;
```



PK plots - concentration curve with regression line

```
title 'Concentration curve with regresson line (Tmax-Tlast) ID=1';
proc sgpanel data=kel noautolegend ;
where id = 1 AND t GE tmax;
panelby  trt / novarname layout=panel columns=1 ;
series x=t y=reg/ lineattrs=(color=black pattern=1);
scatter x = t y=concen
/ markerattrs=(symbol=circle);
scatter x = t y=concen2
/ markerattrs=(symbol=circlefilled);
rowaxis type=log logbase=e;
run;
```



PK plots – Bioequivalence ratio

```
proc sgplot data=ratio_est;
title 'Estimated geometric mean ratios (Test/Reference) ';
scatter Y=ratio X=param / yerrorlower=lower yerrorupper=upper;
yaxis values=(0.60  to 1.60 by 0.10) label= 'Ratio';
xaxis display=(nolabel);
refline 1.25 0.8 /axis=y lineattrs=(color=gray pattern=2);
run;
```



PK plots – Bioequivalence ratio + individual ratios

```
data ratios;
set ratio_est ind_ratios;
if i_ratio EQ . AND param EQ 'AUCt' then x_all = 2;
if i_ratio NE . AND param EQ 'AUCt' then x_all2 = 2.2;
if i_ratio EQ . AND param EQ 'AUCinf' then x_all = 6;
if i_ratio NE . AND param EQ 'AUCinf' then x_all2 = 6.2;
if i_ratio EQ . AND param EQ 'Cmax' then x_all = 10;
if i_ratio NE . AND param EQ 'Cmax' then x_all2 = 10.2;
run;
```



PK plots – Bioequivalence ratio + individual ratios

```
proc sgplot data=ratios noautolegend;
title 'Estimated geometric mean ratios (Test/Reference) ';
scatter Y=ratio X=x_all / yerrorlower=lower yerrorupper=upper
markerattrs=(color=black symbol=circle);
scatter Y=i_ratio X=x_all2 / markerattrs=(color=gray symbol=circle size=3);
yaxis values=(0.70  to 1.70 by 0.10) label= 'Ratio';
xaxis  display= (nolabel noline noticks) tickvalueformat= pkparam. ;
refline 1.25 0.8 /axis=y lineattrs=(color=gray pattern=2);
run;
```



PK plots – Bioeqivalence of highly variable drug (ladder plot)

*EMA (*CPMP/EWP/QWP/1401/98 Rev. 1/ Corr **):Highly variable drug products are those whose intra-subject variability for a parameter is larger than 30%.* ***The applicant should justify that the calculated intra-subject variability is a reliable estimate and that it is not the result of outliers.***



PK plots – ladder plot

```
data graph1;
set pk;
if seq_nro EQ 1 then X1=seq_nro+visit-1;
if seq_nro EQ 2 then X2=seq_nro+visit+2;
if seq_nro EQ 3 then X3=seq_nro+visit+5;
x_all = sum(x1,x2,x3);
l_cmax = log(Cmax);
run;
```




```
title 'Ladder plot by sequencies';
proc sgplot data=graph1 noautolegend;
series y=l_cmax x=x_all  / group=id markers
lineattrs = (pattern=1 thickness = 0.5 color=black)
markerattrs= (symbol=circle color=black size=5);
xaxis values=(0 to 12 by 1)  display= (nolabel noline noticks)
   tickvalueformat= cvar. ;
yaxis label='logCmax' values=(2 to 5 by 0.5);
run;
```